unittest
{
	assert( [ "7 1 3 6 7", ".#..#.." ].parse.expand.solve == "Yes" );
	assert( [ "7 1 3 7 6", ".#..#.." ].parse.expand.solve == "No" );
	assert( [ "15 1 3 15 13", "...#.#...#.#..." ].parse.expand.solve == "Yes" );

	assert( [ "7 1 3 7 5", ".#.#..." ].parse.expand.solve == "No" );		// 追加ケース
	assert( [ "7 1 3 7 6", ".#.#..." ].parse.expand.solve == "Yes" );		// 追加ケース
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nabcd = input.front.split.to!( long[] );
	input.popFront;
	auto s = input.front.to!( char[] );
	return tuple( nabcd[ 0 ], nabcd[ 1 ], nabcd[ 2 ], nabcd[ 3 ], nabcd[ 4 ], s );
}

auto solve( long n, long a, long b, long c, long d, char[] s )
{
	// 移動範囲内に飛び越せない二連岩があったらダメ
	foreach( g; s[ a .. max( c, d ) ].group )
	{
		if( g[ 0 ] == '#' && 2 <= g[ 1 ] ) return "No";
	}
	
	// すぬけ君がふぬけ君を飛び越える必要がなければイイ
	if( c < d ) return "Yes";
	
	// すぬけ君がふぬけ君を飛び越えるための三連スペースがあったらイイ
	foreach( g; s[ b .. d + 1 ].group )
	{
		if( g[ 0 ] == '.' && 3 <= g[ 1 ] ) return "Yes";
	}
	return "No";
}
